
Abstract
In this paper, we revisit the theory of first-class substitution in
contextual type theory (CTT); in particular, we focus on the abstract
notion of substitution variables. This forms the basis for extending
Beluga, a dependently typed proof and programming language
which already supports first-class contexts and contextual objects,
with first-class substitutions. To illustrate the elegance and power of
first-class substitution variables, we describe the implementation of
a weak normalization proof for the simply-typed lambda-calculus
in Beluga.

1. Introduction
Substitutions are a common key concept in describing and reason-
ing about logics and formal systems in general. In this setting, a
substitution describes the mapping between variables of some type
A to objects of the same type. In programming languages, substi-
tutions are used to keep track of the binding of variables to values
and are often referred to as environments. This naturally gives rise
to an environment-based operational semantics where we pair an
open expression M together with the environment which provides
the appropriate value bindings for the free variables in M . Envi-
ronments also play a key role when describing program transfor-
mations such as closure conversion. Because the concept of envi-
ronments is pervasive when talking about programming languages,
some programming languages even provide them as first-class ob-
jects giving programmers direct access to inspect and manipulate
the environment. Gelernter et al. [1987] were one of the first to
propose first-class environments in the setting of LISP where en-
vironments can be passed as arguments, analyzed, and returned as
results. Subsequently, first-class environments have been for exam-
ple incorporated into Scheme and proposed as a means for code
sharing (see for example [Queinnec and Roure 1996]).

From a more theoretical perspective, first-class environments
have been studied by Nishizaki [2000], Sato et al. [2001], and
subsequently by Sato et al. [2002] in the simply typed setting.

In this paper, we take a different view and revisit first-class
substitutions in contextual type theory [Nanevski et al. 2008] which
were first proposed in Pientka [2008]. Fundamentally, first-class
substitutions relate two contexts Γ and Ψ by mapping variables of Γ
to object of the same type in the context Ψ. We therefore introduce
the type Γ[Ψ] which classifies substitutions with domain Γ and
range Ψ. One of the simplest forms of first-class substitutions arises

when we want to relate abstractly two contexts via weakening or
exchange. For example, we might have a lemma which establishes
a property P in a context Γ which contains natural numbers. We
now want to use the result of the lemma in a context Ψ which
contains natural numbers and equality assumptions. To use P in
the context Ψ, we must weaken it.

When reasoning about formal systems, substitutions also natu-
rally arise for example in logical relations proofs such as normal-
ization proofs. In the fundamental lemma of a normalization proof,
we typically prove that given a well-typed term M in a context Γ
and an environment ρ which provides reducible terms for all vari-
ables in Γ, that [ρ]M is reducible.

As mentioned, first-class substitutions were introduced in Pien-
tka [2008]; however there were several issues left open and unsat-
isfactory: what role can substitution variables play in the practice
of mechanizing the meta-theory of formal systems? How do substi-
tution variables compose with identity substitutions? How do sub-
stitution variables compose with variables? What is the canonical
form of a substitution? How do we instantiate and replace substitu-
tion variables?

In this work, we develop a full theory of substitution variables
within contextual type theory CTTs. Our theory takes a fresh look
at the definition of substitutions and enforces a canonical structure
for substitutions, i.e. there is exactly one way to write a substi-
tution. We prove various substitution properties and give a decid-
able, bi-directional type system for our theory. We also have mecha-
nized these properties restricted to the simply-typed setting in Agda
which increases our confidence in the correctness.

CTT extended with substitution variables can also be plugged
into Beluga [Pientka and Dunfield 2010], a dependently-typed pro-
gramming and proof environment which already supports abstract-
ing over and pattern matching on contexts and contextual objects
thereby allowing programmers to abstract over substitutions, to
pass them as arguments and return them as results. We illustrate the
elegance of first-class substitutions by presenting the implementa-
tion of a weak normalization proof for the simply-typed lambda-
calculus. It directly mirrors the theoretical development on paper
which we find a remarkably elegant solution. More generally, any
logical relations proof (parametricity, completeness proofs, ade-
quacy and full abstraction) benefits from our direct, first-class treat-
ment of substitutions.

2. Example: Weak Normalization
In this section we present an example illustrating the application of
substitution variables in a weak normalization proof.

2.1 Hand written proof
We begin by illustrating a typical logical relations proof of weak
normalization for a simply typed lambda calculus to recap the
encoding of formal systems and proofs in Beluga, and to emphasize
the applications of a first-class notion of substitution.

First-class substitutions in contextual type theory 1 2013/6/17



The grammar is straightforward: it includes lambda abstrac-
tions, variables, application, and a single constant of a constant type
i. Lambda abstractions and the constant c are considered values.

Types A,B ::= i | A→ B
Terms M,N ::= c | x | λx.M |M N
Values V ::= c | λx.M

The typing rules are standard and use the judgment Γ `M : A
where Γ describes a typing context.

Γ(x) = A

Γ ` x : A

Γ, x:A `M : B

Γ ` λx.M : A→ B

Γ ` c : i
Γ `M : A→ B Γ ` N : A

Γ `M N : B

We then define a form of weak head reduction with transitivity
and reflexivity built in, simply because it is compact and convenient
for this proof.

M −→M
refl

M −→M ′ M ′ −→ N
M −→ N

trans

(λx.M)N −→ [N/x]M
beta

M −→M ′

M N −→M ′ N
app

We say that a term M halts if there exists a value V such that
M −→ V .

We can now define the notion of reducibility which is typical for
logical relations proofs. A term is reducible at base type precisely
when it halts. A term M is reducible at arrow type A→ B when it
halts, and for every reducible N at type A, the application of M to
N is reducible at type B.

Ri = {M |M halts}
RA→B = {M |M halts and ∀N ∈ RA, (M N) ∈ RB}

It is trivial from the definition that if a term M is reducible at
some type A, then it halts. One can easily prove that RA is closed
under expansion by induction on the type:

Lemma 2.1 (Closure under expansion). If M ′ ∈ RA and M −→
M ′ then M ∈ RA

Our aim is to show that all well-typed closed terms are re-
ducible, but as usual we must generalize to showing that all closed
instantiations of a well-typed open term are reducible. For this we
need the notion of a simultaneous substitution, and to define re-
ducibility of substitutions. If σ = M1/x1,M2/x2, ...,Mn/xn, we
say σ is reducible at context Γ = x1:A1, ..., xn:An if each Mi is
reducible at Ai, i.e. Mi ∈ RAi . We write this as σ ∈ RΓ. We can
now state the main lemma:

Lemma 2.2 (Main lemma). If Γ ` M : A and σ ∈ RΓ then
[σ]M ∈ RA.

Proof. By induction on the typing derivation. We show only the
interesting case:

Case
Γ, x:A `M : B

Γ ` λx.M : A→ B
:

First, [σ](λx.M) = λx.([σ, x/x]M) halts, since it is a value.
Suppose then that we are given N ∈ RA.

1. [σ,N/x]M ∈ RB (by I.H.)
2. [N/x][σ, x/x]M ∈ RB (properties of substitution)
3. (λx.([σ, x/x]M)) N ∈ RB (by closure under expansion)

Hence [σ](λx.M) ∈ RA→B (by definition)

Corollary 2.3 (Weak normalization). If `M : A then M halts.

Proof. By the main lemma, we have that [·]M ∈ RA and hence
that [·]M(= M) halts.

We wish to draw attention to the use of properties of simul-
taneous substitutions in the main lemma and its corollary. These
properties are typically the source of the most overhead when for-
malizing results using various low level representations of variables
and variable binding. However, we will illustrate that these proper-
ties are immediate in our framework, as the equational theory of
(simultaneous) substitutions is built into our type theory.

2.2 Encoding in Beluga
Here we demonstrate the particularly elegant encoding of this proof
in Beluga using first class (simultaneous) substitutions. By hand
we defined grammar and typing separately, but here it is actually
more convenient to define intrinsically typed terms directly. Below,
tm defines our family of simply-typed lambda terms indexed by
their type as an LF signature. In typical higher-order abstract syn-
tax fashion, lambda abstraction takes a function representing the
abstraction of a term over a variable. There is no case for variables,
as they are treated implicitly. We remind the reader that this is a
weak, representational function space – there is no case analysis
or recursion, and hence only genuine lambda terms can be repre-
sented.

tp : type.
i : tp.
arr : tp → tp → tp.

tm : tp → type.
app : tm (arr A B) → tm A → tm B.
lam : (tm A → tm B) → tm (arr A B).
c : tm i.

We can then encode our step relation in a similar fashion below.
Notice in particular we use LF’s notion of substitution to encode
the object-level substitution in the s/beta case.

mstep : tm A → tm A → type.
s/beta : mstep (app (lam M) N) (M N).
s/app : mstep M M’ → mstep (app M N) (app M’ N).
s/refl : mstep M M.
s/trans: mstep M N → mstep N M’ → mstep M M’.

We define a predicate on terms expressing what it means to be a
value: again, the constant c and lambda abstractions are our values.

val : tm A → type.
val/c : val c.
val/lam : val (lam M).

A term halts if it reduces to a value. In the following, M’ is im-
plicitly universally quantified by the constructor h/val, and so this
has the effect of encoding the existential statement “There exists a
value M ′ that M steps to”. As in Twelf Pfenning and Schürmann
[1999], type reconstruction infers types for any free variable in a
given type or kind declaration and implicitly quantifies over them.
Programmers subsequently do not need to supply arguments for
implicitly quantified variables.

halts : tm A → type.
h/val : mstep M M’ → val M’ → halts M.

Reducibility cannot be directly encoded at the LF layer, since it
involves a strong, computational function space. Hence we move to
the computation layer of Beluga, and employ an indexed recursive
type, as defined by Cave and Pientka [2012]. Contextual LF objects
and contexts which are embedded into computation-level types and
programs are written inside [ ].

Once again, a term of type i is reducible if it halts, and a term
M of type arr A B is reducible if it halts, and moreover for every
reducible N of type A, the application app M N is reducible. We
write {N:[.tm A]} for explicit Π-quantification over N, a closed
term of type A. To the left of the dot in [.tm A] is where one writes
the context the term is defined in – in this case, it is empty.

datatype Reduce : {A:[.tp]} {M:[.tm A]} ctype =
| I : [. halts M] → Reduce [. i] [. M]
| Arr : [. halts M] →

({N:[.tm A]} Reduce [. A] [. N]
→ Reduce [. B] [. app M N])

→ Reduce [. arr A B] [. M];

In this definition, the arrows represent the usual computational
function space, not the weak function space of LF. We note that
this definition is not (strictly) positive, since Reduce appears to
the left of an arrow in the Arr case. Allowing unrestricted such
definitions destroys the soundness of our system. Here we note that
this definition is stratified by the type: the recursive occurences of
Reduce are at types A and B which are smaller than arr A B. This
is how one justifies its existence. Beluga does not currently perform
either a positivity check or a stratification check, so in the present
we must convince ourselves that this predicate is well-defined.

The next step we elided in the handwritten proof. It is a trivial
lemma showing that halts is closed under expansion. Proofs in Bel-
uga are written as recursive functions, analyzing the construction of
the underlying LF objects, making recursive calls for appeals to the
induction hypothesis. Here, we proceed by unboxing and pattern
matching on our arguments to extract the witness of the existential
and rebuilding them, appealing to transitivity of mstep.

rec haltsClosed : [. mstep M M’] → [. halts M’]
→ [. halts M] =

fn ms ⇒ fn h ⇒
let [. h/val MS’ V] = h in
let [. MS] = ms in
[. h/val (trans MS MS’) V];

The trivial fact that reducible terms halt has a corresponding
trivial proof, analyzing the construction of the the proof of Reduce
[.A] [.M]:

rec cr1 : Reduce [.A] [.M] → [. halts M] =
fn r ⇒ case r of
| I h ⇒ h
| Arr h f ⇒ h;

Next we prove closure of Reduce under expansion. This fol-
lows the handwritten proof, proceeding by induction on the type
A which is an implicit argument. In the base case we appeal to
haltsClosed, while in the Arr case we must also appeal to the
induction hypothesis at the range type, going inside the function
position of applications.

rec closed : [. mstep M M’] → Reduce [.A] [.M’]
→ Reduce [.A] [.M] =

fn ms ⇒ fn r ⇒ case r of
| I h ⇒ I (haltsClosed ms h)
| Arr h f ⇒ let [.MS] = ms in

Arr (haltsClosed ms h)
(λ N ⇒ fn rn ⇒
closed [. s/app MS] (f [. N] rn));

Note that we overload λ; here we use λ-abstractions when
abstracting over an index argument explicitely, while we use fn-
abstraction when abstracting over a computation-level variable.
Now we arrive at the part of the proof requiring simultaneous
substitutions. First we define a schema for contexts, describing a
type of contexts which contains only (typed) term assumptions:

schema ctx = tm T;

Schema declarations are similar to world declarations in Twelf
[Pfenning and Schürmann 1999]. In the example above, ctx clas-
sifies contexts which contain instances of tm T; for example, x:tm
i, y: tm (arr i i), w: tm i is a well-formed context of

schema ctx.
We now must state precisely what it means for a substitution to

be reducible. We do this by employing another indexed recursive
type: a predicate expressing that the substitution was built up as
a list of reducible terms. The notation #S stands for a substitution
variable. Its type is written g[], meaning that it has domain g and
empty range, i.e. it takes variables in g to closed terms of the same
type. In the base case, the empty substitution is reducible. In the
Cons case, we read this as saying: if #S is a reducible substitution
(implicitly at type g[]) and M is a reducible term at type A, then
#S with M appended is a reducible substitution (implicitly at type
(g,x:tm A)[] – the domain has been extended with a variable of
type A).

datatype RedSub : (g:ctx){#S:g[]} ctype =
| Nil : RedSub [. · ]
| Cons : RedSub [. #S] → Reduce [.A] [.M] →

RedSub [. #S M ];

We implicitly quantify over the context g by using round paren-
thesis writing (g:ctx); we explictly quantify over substitution
variables using curly braces writing {#S:g[]}.

The main fact that we need about this definition is that if a sub-
stitution #S is reducible, then looking up a variable #p in it pro-
duces something reducible. The hash in front of #p distinguishes it
as a parameter variable, i.e. standing for an object-level variable.
To clarify, #p #S is notation for applying the substitution #S to the
variable #p – informally, σ(x). The proof proceeds by analyzing
the variable. If it is the top variable of the context, the proof that
#p #S is reducible is the one at the top. Otherwise, we continue in-
ductively looking in the prefix of the context. The case of an empty
context is impossible, as there are no variables in the empty context.

rec lookup : {g:ctx}{#p:[g.tm A]} RedSub [. #S] →
Reduce [.T] [. #p #S] =

λg ⇒ λ#p ⇒ fn rs ⇒
case [g. #p ..] of
| [g’,x:tm A. x] ⇒ let Cons rs’ rN = rs in rN
| [g’,x:tm A. #q ..] ⇒ let Cons rs’ rN = rs in

lookup [g’] [g’. #q ..] rs’;

In the above, we write .. for the identity substitution associated
with a context variable. In particular, when we write [g’,x:tm A.
#q ..], this means that #q is a parameter variable defined in g’,

and hence is not x. If we wished to allow it to be x, we would
instead write #q .. x, where now .. x is the expanded identity
substitution for the context g’,x:tm A. One may choose to think
of the .. in this instance as a shift. Operationally, when we have a
concrete variable for #p, this pattern performs a check to see if the
variable is in the image of the .. pattern substitution, i.e. if it is not
the top variable x. One may refer to the work of ? for details on the
operational interpretation of these patterns.

Finally, our main lemma is standard and takes the form we
would expect from the handwritten proof: if M is a well-typed term,
and we provide a reducible substitution #S with closed instantia-
tions for each of the free variables of M, then M #S (that is, the
application of #S to M) is reducible. We proceed by induction on
the term. When it is a variable, we appeal to lookup. When it is
an application, we straightforwardly apply the functional argument
we obtain from the induction hypothesis for M1 to the induction
hypothesis for M2. The lam case is the most interesting. Clearly the
lambda abstraction halts, as it is a value. To show that applying it to

any reducible term is reducible, we appeal directly to closure under
expansion, and the induction hypothesis for M1 with the substitu-
tion extended with N. Recall that rN is a proof that N is reducible.
We write _ for the type of the variable x and let reconstruction
infer it.

rec main : {g:ctx}{M:[g.tm A]} RedSub [. #S]
→ Reduce [.A] [. M #S] =

λg ⇒ λM ⇒ fn rs ⇒ case [g. M ..] of
| [g. #p ..] ⇒ lookup [g] [g. #p ..] rs

| [g. lam (λx. M1 .. x)] ⇒
Arr [. h/val s/refl val/lam]
(λ N ⇒ fn rN ⇒ closed [. s/beta]
(main [g,x:tm _] [g,x. M1 .. x] (Cons rs rN)))

| [g. app (M1 ..) (M2 ..)] ⇒
let Arr h f = main [g] [g. M1 ..] rs in
f [._] (main [g] [g. M2 ..] rs)

| [g. c] ⇒ I [. h/val s/refl val/c];

Notably, we did not have to concern ourselves with the property
of substitutions that we wrote explicitly in the paper proof in the
lam case. Using a low-level representation of substitution (e.g. de
Bruijn indices) one must prove [σ,N/x]M = [N/x][σ, x/x]M
(where x 6∈ FV(σ)) by hand, and this is actually a large bulk of the
proof. We will see that in our setting, it is handled automatically by
normalization during typechecking: the normal form of both sides
of this equation is [σ,N/x]M .

Weak normalization is now a trivial corollary, taking M to be a
closed term and #S to be the empty substitution:

rec weakNorm : {M:[. tm A]} [. halts M] =
λM ⇒ cr1 (main [] [. M] Nil);

Once again if one is being pedantic (as one must in a formaliza-
tion), this requires the property that [·]M = M , which is arranged
to hold automatically in our system.

We wish to point out that the whole development is ≈60 lines,
and it follows the handwritten proof very closely, with essentially
no extra overhead. Compared to low-level techniques for variable
binding, it is a huge win to not be burdened with proving proper-
ties of simultaneous substitution. Even in other systems employing
higher-order abstract syntax, such as Abella [Gacek 2008], simul-
taneous substitution is not a first-class notion, and must be defined
explicitly. This means that one must still prove properties of simul-
taneous substitution, although it is somewhat easier in Abella than
with a low-level representation because simultaneous substitution
can be built out of the provided individual substitution. However,
one must still explicitly prove that the defined simultaneous substi-
tution is a congruence in Abella, i.e. ([σ]M) ([σ]N) = [σ](M N)
and similarly for the more complex λ-abstraction case. In our sys-
tem, this burden is lifted completely.

For systems such as Twelf [Pfenning and Schürmann 1999] and
Delphin [Poswolsky and Schürmann 2008] a direct formulation of
normalization proofs is out of reach, since they lack first-class con-
texts and recursive types. Instead Schürmann and Sarnat [2008]
proposed to represent and reason about an auxiliary logic to over-
come the limited meta-logical strength of systems such as Twelf.

Preliminary results suggest that this proof can be elegantly
extended to a βη-normalization proof which normalizes under
binders. This proof is made substantially more difficult by the
fact that one must invent an arbitrary number of fresh variables
when η-expanding. A common technique is to employ a Kripke
logical relation in which one quantifies over a variable-for-variable
substitution representing an arbitrary weakening. This proof relies
much more heavily on the equational theory of substitutions, so

our approach is particularly promising here. More generally, our
approach should prove useful for many (Kripke) logical relations
proofs, such as parametricity, full abstraction, or various kinds of
completeness proofs. This is especially so for larger languages
(e.g. with case expressions) where one must use such properties of
substitution repeatedly.

3. Contextual LF with Substitution Variables
Here we revisit contextual LF [Nanevski et al. 2008; Pientka 2008]
which extends the logical framework LF [Harper et al. 1993] with
contextual objects and types. A contextual type A[Ψ] characterizes
a LF object M of type A in a context Ψ; it is inhabited by the
contextual object [Ψ̂.M ] where Ψ̂ lists the free variables occurring
in M and can be obtained from the context Ψ by erasing the
type annotations and keeping only the declared variable names.
In Pientka [2008], contextual LF was extended to allow context
variables which inhabit a context schema and parameter variables
which inhabit the parameter type #A[Ψ]. In this work, we revisit
the addition of first-class substitutions and substitution variables.

3.1 Contextual LF
We summarize here contextual LF concentrating on the new parts
which deal with substitution variables. Furthermore, we concen-
trate here on characterizing well-typed terms, but defining kinds
and kinding rules for types is straightforward and omitted. We char-
acterize only objects in βη normal form, since these are the only
meaningful objects in LF.

Atomic types P,Q ::= a · S
Types A,B ::= P | Πx:A.B

Heads H ::= x | c | p[σ] | (Ψ̂.x)[ρ]
Spines S ::= ε | N ;S
Neutral Terms R ::= H · S | u[σ]
Normal Terms M,N ::= R | λx.M
Substitutions σ, τ ::= · | idψ | σ,M | idψ[ρ]
Sub. Closures ρ ::= s[σ]
Contexts Ψ ::= · | ψ | Ψ, x:A

We present the language in spine form [Cervesato and Pfenning
2003], as it makes the termination of hereditary substitution easier
to establish. Instead of x M1 . . .Mn, we write x ·M1; . . . ;Mn; ε.

In addition to λ-abstraction, application, variables and con-
stants, contextual LF also contains meta-variables u which rep-
resent a general open LF object and parameter variables p which
represents an LF variable. They are associated with a postponed
substitution σ representing a closure. The intention is to apply σ as
soon as we know what u (or p resp.) stand for. Intuitively, we can
justify the closure as follows: let u describe a hole in a term M ; if
we were to apply a substitution σ to M , then σ gets stuck when it
encounters the hole u. However, as soon as we know what u stands
for, we can further apply the substitution σ to it.

Meta-variables are considered neutral terms and our typing rules
will enforce that meta-variables are of atomic type. This is not a re-
striction, since we can always lower a meta-variable; intuitively, we
transform u[σ]M (or in our syntax u[σ] ·M ; ε) where u has type
(Πx:A.B)[Ψ] to u′[σ,M ] where u′ has type B[Ψ, x:A] by replac-
ing u with Ψ̂.λx.u′[σ, x] possibly η-expanding x, if necessary.

For this paper, we draw the reader’s attention to substitution
variables s. They appear in two forms in the grammar. If the domain
of a substitution variable s is Ψ and x is a variable in Ψ then looking
up the variable x in s (i.e. s(x)) is stuck until one has a concrete
instantiation for s. Moreover, applying a substitution σ to s(x) is
also stuck until one has a concrete instantiation for s. This is the
meaning of the term (Ψ̂.x)[s[σ]]. One may also view the domain
of a substitution as a form of product type, in which case the term

(Ψ̂.x)[s[σ]] may be read as the x projection of s[σ]. The Ψ̂ binds
the x, allowing for α-renaming.

The second form, idψ[s[σ]] is morally the stuck composition
[σ]([s](idψ)), where idψ may be thought of as a weakening or shift-
ing substitution. If the domain of s starts with a context variable
(e.g. ψ,Ψ) then this corresponds to removing the Ψ components,
leaving only the ψ components of the substitution.

The grammar and typing rules ensure that substitutions are
η-long. By this we mean that if a substitution closure ρ (=
s[σ]) has domain ψ, x:A, then one may write the substitution
(idψ[ρ], (ψ,x. x)[ρ]) which morally represents the same thing as
ρ, but the term ρ by itself is not even grammatically a substitu-
tion. This has the practical impact that when writing Beluga pro-
grams, one never needs to do work to prove an equation such as
ρ = idψ[ρ], (ψ,x. x)[ρ], since it is arranged to hold in the theory.
Similarly, if ρ has domain ·, the η-long form of ρ is simply ·, the
empty substitution. Note that our treatment of substitution variables
differs here from the treatment in [Pientka 2008] where we allow
s[σ] to stand by itself.

Next we define hereditary substitution, which performs a substi-
tution and reduces any redeces which arise along the way. Hered-
itary substitution allows us to only consider normal forms in our
grammar and typing rules, which makes the decidability of type-
checking obvious. As usual, we annotate hereditary substitutions
with an approximation of the type of the term we substitute for to
guarantee termination.

Type approximations α, β ::= a | α→ β

We then define the dependency erasure operator (−)− as fol-
lows:

(a · S)− = a
(Πx:A.B)− = (A)− → (B)−

We will sometimes tacitly apply the dependency erasure oper-
ator (−)− in the following definitions. We can now define hered-
itary substitution, presented in Figure 1. The definition is mostly
standard. As usual, when the substitution would create redeces,
we proceed by hereditarily performing substitution. The addition
of substitution variables does not pose any difficulties here. Termi-
nation is readily established:

Lemma 3.1. The hereditary substitutions [M/x]∗α(N) where ∗ ∈
{n, s, l} and reduce(M : α, S) terminate, either by failing or
successfully producing a result.

Proof. By nested induction on α primarily and N secondarily.

Now we can generalize hereditary substitution to simultaneous
substitutions. It is this generalization (and later, meta-substitution)
which needs a careful treatment in the presence of substitution
variables. Note that previous formulations of contextual LF allow
substitutions to be extended with variables in addition to arbitrary
terms. This is motivated by the fact that as we push σ under a
binder in λx.M , we would like to extend it by the identity mapping
x to itself. However, since we do not know the type of x, simply
extending σ with x may not be meaningful, since x may not be in
η-long form. Therefore, previous formulations allow a substitution
σ to be extended with M where M is either an η-long normal
form or a variable. Unfortunately, this complicated the grammar
of substitutions and meant that there was not necessarily a unique
representation of substitutions. This meant that substitutions still
had to be compared up to η equality. Our formulation avoids these
issues and requires that substitutions must be extended with η-long
terms, thus guaranteeing unique normal forms for substitutions.

Normal terms

[M/x]nα(λy.N) = λy.N ′ where N ′ = [M/x]nα(N)
choosing y 6∈ FV(M)
and y 6= x

[M/x]nα(u[σ]) = u[σ′] where σ′ = [M/x]sα(σ)

[M/x]nα(c · S) = c · S′ where S′ = [M/x]lαS
[M/x]nα(x · S) = M ′ where S′ = [M/x]lα(S)

and M ′ =
reduce(M :α, S′)

[M/x]nα(y · S) = y · S′ where y 6= x
and S′ = [M/x]lα(S)

[M/x]nα(p[σ] · S) = p[σ′] · S′ where σ′ = [M/x]sα(σ)

[M/x]nα((Ψ̂.y)[ρ] · S) = (Ψ̂.y)[ρ′] · S′ where ρ′ = [M/x]α(ρ)
and S′ = [M/x]lα(S)

Spines

[M/x]lα(ε) = ε
[M/x]lα(N ;S) = N ′;S′ where N ′ = [M/x]nα(N)

and S′ = [M/x]lα(S)
Substitutions

[M/x]sα(·) = ·
[M/x]sα(idψ) = idψ
[M/x]sα(σ,N) = σ′, N ′ where σ′ = [M/x]sα(σ)

and N ′ = [M/x]sα(N)
[M/x]sα(idψ[ρ]) = idψ[ρ′] where ρ′ = [M/x]α(ρ)

Substitution Closures

[M/x]α(s[σ]) = s[σ′] where [M/x]α(σ) = σ′

reduce(λx.M : α→ β, (N ;S)) = reduce([N/x]nα(M) : β, S)
reduce(R : a, ε) = R
reduce(M,S) fails otherwise

Figure 1. Hereditary substitution

In Figure 2 we define an operation [σ]Φ̂Ψ. This operation is a
true simultaneous substitution when Φ̂ = ·. More generally it
means substitute for the variables in Ψ, leaving the variables in Φ̂
untouched. It is a simultaneous substitution under a context. This
means that when we go under the binder in λx.M we need only to
extend Φ̂ with x. This is the key to avoiding the issues mentioned
above. The typing lemma below should shed some light on the
behaviour of this operation.

Since the domain of our substitutions is not explicit in the
syntax, in the application [σ]Ψ(M) we consider the ith component
of σ to correspond to the ith component of Ψ, thus order is relevant.
We will write [σ]Ψ(−) to mean [σ]·Ψ(−), dropping the dot.

We highlight here a few key parts of this definition. Because
substitutions are η-long, variable lookup is guaranteed to succeed
if the variable is in the domain of the substitution. i.e. the case
(s[σ])Ψ,x:A(x) which is problematic in Pientka [2008] cannot arise
here. The truncation operation truncΨ(σ) has the effect of dropping
components of σ, keeping only the portion supplying instantiations
for a context variable. This arises when applying [σ]Φ̂Ψ(idψ). From
the typing, we will know that σ must begin with a substitution hav-
ing domain ψ and that this operation must produce a substitution
with domain ψ. So we simply chop off the suffix we do not need.
That is, idψ acts as a weakening substitution. Since σ must be in

Variable lookup

(σ,M)Ψ,x:A(x) = M : (A)−

(σ,M)Ψ,y:A(x) = σΨ(x) where x 6= y
(σ)Ψ(x) fails otherwise

Truncation

truncψ(σ) = σ
truncΨ,x:A(σ,N) = truncΨ(σ)
truncΨ(σ) fails otherwise

Normal terms

[σ]Φ̂Ψ(λy.N) = λy.N ′ where N ′ = [σ]Φ̂,yΨ (N)
choosing y 6∈ FV(σ)
and y 6∈ Ψ

[σ]Φ̂Ψ(u[τ ]) = u[τ ′] where τ ′ = [σ]Φ̂Ψ(τ)

[σ]Φ̂Ψ(c · S) = c · S′ where S′ = [σ]Φ̂ΨS

[σ]Φ̂Ψ(x · S) = reduce(M : α, S′) where x 6∈ Φ̂
and σΨ(x) = M : α

and S′ = [σ]Φ̂Ψ(S)

[σ]Φ̂Ψ(y · S) = y · S′ where y ∈ Φ̂

and S′ = [σ]Φ̂Ψ(S)

[σ]Φ̂Ψ(p[τ ] · S) = p[τ ′] · S′ where τ ′ = [σ]Φ̂Ψ(τ)

[σ]Φ̂Ψ((Ψ̂′.x)[ρ] · S) = (Ψ̂′.x)[ρ′] · S′ where ρ = [σ]Φ̂Ψ(ρ)

and S′ = [σ]Φ̂Ψ(S)
Spines

[σ]Φ̂Ψ(ε) = ε

[σ]Φ̂Ψ(N ;S) = N ′;S′ where N ′ = [σ]Φ̂Ψ(N)

and S′ = [σ]Φ̂Ψ(S)
Substitutions

[σ]Φ̂Ψ(·) = ·
[σ]Φ̂Ψ(idψ) = truncΨ(σ)

[σ]Φ̂Ψ(τ,N) = τ ′, N ′ where τ ′ = [σ]Φ̂Ψ(τ)

and N ′ = [σ]Φ̂Ψ(N)

[σ]Φ̂Ψ(idψ[ρ]) = idψ[ρ′] where ρ′ = [σ]Φ̂Ψ(ρ)

Substitution Closures

[σ]Φ̂Ψ(s[τ ]) = s[τ ′] where τ ′ = [σ]Φ̂Ψ(τ)

Figure 2. Simultaneous hereditary substitution

fully expanded form, the problematic case of truncΨ,x:A(s[σ]) in
[Pientka 2008] is not possible here.

There is one last tool we need to define the typing rules. We
will need certain expansion operations. In particular, we need to
compute the expanded form of s[σ] at domain Ψ, which in turn
also requires η-expansion. We also define here the related operation
of expanding identity substitutions. We will make more extensive
use of these operations in the next section. Below we define η-
expansion on the approximate type, using @ to mean appending
a term on the end of a spine. In a spine calculus, η-expansion takes
a head H and a spine S, and computes the η-long form of H · S:

η-expa(H,S) = H · S
η-expα→β(H,S) = λx.η-expβ(H, (S@η-expα(x)))

where x 6∈ FV(H) ∪ FV(S)

We write η-expα(H) as shorthand for η-expα(H, ε). In the
arrow case, we introduce a fresh variable x and continue expanding
H applied to the spine S extended with x. We can now expand the
identity substitution for a context Ψ by appealing to η-expansion.
We simply walk through the context, η-expanding each variable.
When we encounter a context variable, we are stuck (that is, until
we have a concrete instantiation for it), so we make use of the stuck
syntactic form idψ:

id(·) = ·
id(ψ) = idψ
id(Ψ, x:A) = id(Ψ), η-expA(x)

Next we define substitution expansion. The expansion sexpΨ(ρ)
computes the expanded form of ρ when the domain of s is a
general context Ψ. To do so, we must generalize this operation to
perform expansion underneath a context Φ̂. We similarly traverse
the context, for each variable x in the domain Ψ we add the x
projection of ρ. When we encounter a context variable, we are
similarly stuck, and so we use the stuck syntactic form idψ[ρ]:

sexpΦ̂
· (ρ) = ·

sexpΦ̂
ψ(ρ) = idψ[ρ]

sexpΦ̂
Ψ,x:A(ρ) = sexpx,Φ̂Ψ (ρ), η-expA((Ψ̂,x,Φ̂. x)[ρ])

We now have the tools to define the typing of Contextual LF
with substitution variables. We use the following judgements:

∆; Ψ `M ⇐ A Normal term M checks against type A
∆; Ψ ` H ⇒ A Head H synthesizes type A
∆; Ψ ` S > A⇒ B Spine S synthesizes type B
∆; Ψ ` σ ⇐ Φ Substitution σ has domain Φ and range Ψ
∆; Ψ ` ρ⇒ Φ Sub. Closure ρ has domain Φ and range Ψ

The bi-directional typing rules are mostly straightforward and
are presented in Figure 3. The context ∆ contains the various
forms of meta-level variables and is described in more detail in
the next section, while the context Ψ contains regular LF variables.
We will tacitly rename bound variables, and maintain that contexts
and substitutions declare no variable more than once. Note that
substitutions σ are defined only on ordinary variables x and not
contextual variables. Moreover, we require the usual conditions
on bound variables. For example in the rule for λ-abstraction the
bound variable x must be new and cannot already occur in the
context Ψ. This can be always achieved via α-renaming. Similarly,
in meta-terms we tacitly apply α-renaming. We write Ψ,Φ for
context concatenation. When we write this, we mean to imply that
Φ does not start with a context variable, i.e. Φ is a pure list of the
form x1:A1, ..., xn:An.

We highlight here a few of the interesting rules. The rule for
substitution closures ρ of the form s[σ] types the stuck composi-
tion of a substitution variable with a substitution σ. If s takes Φ
to Ψ′ and σ takes Ψ′ to Ψ, then the stuck composition s[σ] takes
Φ to Ψ. Substitution closures can be used in two ways: the first,
idψ[ρ] projects out the ψ components of ρ. So if ρ provides instan-
tiations for ψ and Φ then we can discard those for Φ, leaving only
the instantiations for ψ. The second, (Φ̂.x)[ρ] allows one to use
the instantiations ρ provides for concrete variables. If ρ provides
instantiations for Φ, among which is a variable x:A, then the type
of the stuck projection (Φ̂.x)[ρ] is [sexpΦ(ρ)]Φ(A), in effect ap-
plying ρ to A. We must expand the substitution, since [−] is only
defined for fully-expanded substitutions. This has the unfortunate
behaviour of potentially performing η expansions which then trig-
ger spurious β reductions during hereditary substitution. This is
sufficient for our purposes, although it could be avoided at the cost
of directly defining an operation [ρ](−).

Heads ∆; Ψ ` H ⇒ A

Ψ(x) = A

∆; Ψ ` x⇒ A

∆(p) = #A[Φ] ∆; Ψ ` σ ⇐ Φ

∆; Ψ ` p[σ]⇒ [σ]ΦA

Σ(c) = A

∆; Ψ ` c⇒ A

∆; Ψ ` ρ⇒ Φ Φ(x) = A

∆; Ψ ` (Φ̂.x)[ρ]⇒ [sexpΦ(ρ)]ΦA

Spines ∆; Ψ ` S > A⇒ B

∆; Ψ `M ⇐ A ∆; Ψ ` S > [M/x]AB ⇒ B′

∆; Ψ `M ;S > Πx:A.B ⇒ B′

∆; Ψ ` ε > A⇒ A

Normal Terms ∆; Ψ `M ⇐ A

∆; Ψ ` H ⇒ A ∆; Ψ ` S > A⇒ P P = Q

∆; Ψ ` H · S ⇐ Q

∆(u) = P [Φ] ∆; Ψ ` σ ⇐ Φ [σ]ΦP = Q

∆; Ψ ` u[σ]⇐ Q

∆; Ψ, x:A `M ⇐ B

∆; Ψ ` λx.M ⇐ Πx:A.B

Substitutions ∆; Ψ ` σ ⇐ Ψ′

∆; Ψ ` · ⇐ · ∆;ψ,Ψ ` idψ ⇐ ψ

∆; Ψ ` σ ⇐ Φ ∆; Ψ `M ⇐ [σ]ΦA

∆; Ψ ` σ,M ⇐ Φ, x:A

∆; Ψ ` ρ⇒ ψ,Φ

∆; Ψ ` idψ[ρ]⇐ ψ

Sub. Closures ∆; Ψ ` ρ⇒ Φ

∆(s) = Φ[Ψ′] ∆; Ψ ` σ ⇐ Ψ′

∆; Ψ ` s[σ]⇒ Φ

Figure 3. Typing for contextual LF with substitution variables

The expected weakening properties hold in our type system. The
following lemmas provide the typing for the expansion operations
and hereditary substitution. Below, we use J to stand for any of the
forms of judgements defined above.

Lemma 3.2 (Expansion lemmas).

1. If ∆; Ψ ` H ⇒ A and ∆; Ψ ` S > A ⇒ B and
η-expB(H,S) = M then ∆; Ψ `M ⇐ B

2. If id(Ψ) = σ then ∆; Ψ ` σ ⇐ Ψ

3. If ∆; Ψ′ ` ρ⇒ Ψ,Φ and sexpΦ̂
Ψ(ρ) = σ then ∆; Ψ′ ` σ ⇐ Ψ

Lemma 3.3 (Substitution lemma).

1. If ∆; Ψ `M ⇐ A and ∆; Ψ, x : A,Φ ` J
then ∆; Ψ, [M/x]A(Φ) ` [M/x]∗A(J) where ∗ ∈ {n, s, l}

2. If ∆; Ψ ` M ⇐ A and ∆; Ψ ` S > A ⇒ B and
reduce(M : (A)−, S) = M ′ then ∆; Ψ `M ′ ⇐ B

Lemma 3.4 (Simultaneous substitution lemma).

1. If ∆; Ψ,Φ ` J and ∆; Ψ′ ` σ ⇐ Ψ

then ∆; Ψ′, [σ]·Ψ(Φ) ` [σ]Φ̂Ψ(J)
2. If ∆; Ψ ` σ ⇐ ψ,Φ then ∆; Ψ ` truncψ,Φ(σ)⇐ ψ

Since hereditary substitution terminates, our typing rules are
clearly syntax directed, and equality is simply syntactic, our typing
rules immediately give rise to an algorithm for typechecking.

Theorem 3.5. Typechecking is decidable.

3.2 Meta-Objects and Meta-types
To give meaning to the substitution variables present in our syn-
tax, we must describe what it means to substitute for one. We also
need to carefully describe the interaction of our substitution vari-
ables with context variables and substitution for context variables.
Indeed, since we intend to use Contextual LF with substitution vari-
ables as an index language in Beluga, we must lift it to a meta-level
where we abstract over all the free variables of a term, and we must
define the meta-substitution operation.

We base our presentation of the meta-level here on the pre-
sentation by Cave and Pientka [2012], highlighting the exten-
sions necessary for substitution variables. We lift contextual LF
objects to meta-types and meta-objects to treat abstraction over
meta-objects uniformly. Meta-objects include contextual objects
written as Ψ̂.R, contexts Ψ, and now also contextual substitutions
written Ψ̂.σ. These are the index objects which can be used to
index computation-level types in Beluga. There are four different
flavours of meta-types: P [Ψ] denotes the type of a meta-variable u
and stands for a general contextual object Ψ̂.R. #A[Ψ] denotes the
type of a parameter variable p and it stands for a variable object, i.e.
either Ψ̂.x or Ψ̂.p[π] where π is a variable substitution. A variable
substitution π is a special case of general substitutions σ; however
unlike p[σ] which can produce a general LF object, p[π] guarantees
we are producing a variable. G describes the schema (i.e. type) of
a context. Last, the type Ψ[Φ] characterizes substitutions with do-
main Ψ and range Φ – they take objects defined in the context Ψ to
objects defined in the context Φ.

The tag # on the type of parameter variables is a simple syn-
tactic device to distinguish between the type of meta-variables and
parameter variables. It does not introduce a subtyping relationship
between the type #A[Ψ] and the type A[Ψ]. The meta-context in
which an LF object appears uniquely determines if X denotes a
meta-variable, parameter variable or context variable. We use the
following convention: if X denotes a meta-variable we usually
write u or v; if it stands for a parameter-variable, we write p; for
context variables we use ψ and for substitution variables we use s.

Context schemas G ::= ∃
−−−→
(x:A).B | G+ ∃

−−−→
(x:A).B

Meta Objects C ::= Ψ̂.R | Ψ | Ψ̂.σ
Meta Types U ::= P [Ψ] | #A[Ψ] | Ψ[Φ] | G
Meta substitutions θ ::= · | θ, C/X
Meta-context ∆ ::= · | ∆, X:U

Context schemas consist of different schema elements ∃
−−−→
(x:A).B

which are built using +. Intuitively, this means a concrete declara-
tion in a context must be an instance of one of the elements spec-
ified in the schema. For example, a context x:exp nat, y:exp bool
will check against the schema ∃T :tp.exp T .

The uniform treatment of meta-terms, called C, and meta-
types, called U , allows us to give a compact definition of meta-
substitutions θ and meta-contexts ∆.

We define the typing rules for meta-objects in Figure 4. These
are unchanged from Cave and Pientka [2012], with the exception
of the added rule for substitution objects.

In Figure 5 we present the definition of meta-substitution, which
substitutes a meta object for a metavariable, parameter variable,
context variable, or substitution variable. We give a single defini-
tion of meta-substitution which can substitute for any of these four
flavours of meta-types. We annotate the meta-substitution with the

Meta Terms ∆ ` C ⇐ U

∆ ` · ⇐ G

∆(ψ) = G

∆ ` ψ ⇐ G

∆ ` Ψ⇐ G

∃
−−−−−→
(x : B′).B ∈ G ∆; Ψ ` σ ⇐

−−−−→
(x:B′) A = [σ]−−−−→

(x:B′)
B

∆ ` Ψ, x:A⇐ G

∆; Ψ ` σ ⇐ Φ

∆ ` Ψ̂.σ ⇐ Φ[Ψ]

∆; Ψ ` R⇐ P

∆ ` Ψ̂.R⇐ P [Ψ]

Ψ(x) = A

∆ ` Ψ̂.x⇐ #A[Ψ]

∆(p) = #A[Φ] ∆; Ψ ` π ⇐ Φ [π]Φ(A) = B

∆ ` Ψ̂.p[π]⇐ #B[Ψ]

Meta-Substitutions ∆ ` θ ⇐ ∆′

∆ ` · ⇐ ·
∆ ` θ ⇐ ∆′ ∆ ` C ⇐ [[θ]]∆′(U)

∆ ` θ, C/X ⇐ ∆′, X:U

Figure 4. Typing for meta-terms

type of the object it is substituting, since we need this information
to pass on to hereditary substitution.

The case for metavariables is standard for CTT. When substitut-
ing a meta-object Ψ̂.R for u in u[σ], we first continue substituting
for occurences of u in σ to obtain σ′. Now that we know more con-
cretely what u is we trigger the suspended substitution [σ′]Ψ(R),
appealing now to hereditary substitution.

The cases of interest in this paper are those involving substitu-
tion variables. When we substitute a concrete τ for s in (Φ̂.x)[s[σ]]·
S, we compose it with the suspended σ, lookup x in the resulting
substitution to obtain a normal term, and reduce the redeces which
arise from its application to S.

Two interesting cases arise when substituting into idψ[s[σ]].
When substituting a concrete τ for s, we must compose it with the
suspended σ and drop the suffix of the substitution, keeping only
the ψ component. When substituting a concrete Ψ for ψ, we must
expand s[σ] into a genuine substitution.

This operation obeys the following typing lemma:

Lemma 3.6 (Meta-substitution lemma).
If ∆ ` C ⇐ U and ∆, X:U,∆′; Ψ ` J
then ∆, [[C/X]]U (∆′); [[C/X]]U (Ψ) ` [[C/X]]U (J)

With meta-substitution properly defined, we can simply plug the
meta level into the computation level framework described in Cave
and Pientka [2012] to obtain the type safety results proven there.
As a result, we obtain a language capable of formalizing the weak
normalization example of Section 2.2.

4. Formalization and Implementation
We have carried out a partial mechanization of the results here in
Agda [Norell 2007], a dependently typed programming language.
Namely, we have implemented simply-typed variants of the defini-
tions and results prior to Section 3.2, using strongly-typed term rep-
resentations – for a reference on this technique, see ?. This follows
work by ? on formalizing hereditary substitution for simple types.
Our implementation in Agda serves to establish the totality (termi-
nation and coverage) of the hereditary substitution operations we
define here, as well as simply-typed approximations of the substi-
tution lemmas and expansion lemmas. Note that it does not attempt

to prove any of the algebraic properties of substitution we expect
to hold, as proving these is still burdensome using this technique.
Nevertheless, this proved to be a valuable tool in assisting to de-
velop the theory, in particular because the on-paper definitions are
traditionally written first as partial functions before proving they
are total on well-typed terms. The strongly-typed term represen-
tation, on the other hand, allowed us to directly define hereditary
substitution as a total function, ruling out cases which are impossi-
ble due to typing.

We have also carried out a preliminary implementation of CTTs

in the Beluga system, which is capable of typechecking the ex-
ample in Section 2.2, save for the fact that type reconstruction in-
volving substitution variables is not presently as powerful as the
example suggests, and we must supply more arguments explicitly.
However, we see no fundamental obstacles to improving the recon-
struction to the point where the example can be typechecked as-is.
We plan to integrate substitution variables more smoothly into type
reconstruction and make a public release in the near future.

5. Related Work
Abadi et al. [1990] briefly consider metavariables (including sub-
stitution variables) in the context of their explicit substitution calcu-
lus, concluding that the η rule (surjective pairing) for substitutions
leads to non-confluence. The lack of confluence is not a concern
in our setting, as we are only concerned with normal forms and not
rewriting. Our work provides a precise analysis of the complex nor-
mal forms Abadi et al. allude to. We conjecture that our hereditary
substitution gives rise to a sound and complete decision procedure
for (a typed variant of) their calculus, although it remains to work
out the details.

Subsequently, substitution variables, sometimes referred to as
first-class environments, have been considered for simply-typed
calculi [Hashimoto and Ohori 2001; Mason 1999; Nishizaki 2000;
Sato et al. 2001, 2002]. Nishizaki [2000] for example extends a
lambda-calculus with explicit substitutions following in the spirit
Abadi et al. [1990] and representing variables via de Bruijn indices.

The present work also bears a similarity to the row polymor-
phism of ?. Row polymorphism provides a notion of record poly-
morphism which admits type inference. It allows one to polymor-
phically quantify over row variables representing a subset of the
fields of a record. Here our context variables fill a role similar to
row variables, and our substitutions fill a role similar to records.
Our (Ψ̂.x)[ρ] construction acts as a field access. Through this lens,
it would seem that our work provides a perspective on η-expansion
and η-long forms for row-polymorphic records. One notable dif-
ference is that we consider our substitution types equivalent under
α-renaming, while this is generally not the case for record fields.
Our setting is also dependently typed instead of polymorphically
typed.

6. Conclusion
We have presented a full theory of substitution variables within
contextual type theory which solves several issues, previous pro-
posals left open: it defines a canonical form for substitutions and
we include the closure of a bound variable with a substitution vari-
able as well as the closure of the identity substitution with a substi-
tution variable. Our resulting type theory characterizes only normal
forms and type checking is decidable.

We also illustrate the elegance and usefulness of substitution
variables, when implementing normalization proofs. A preliminary
prototype for programming with first-class substitution variables is
under way within the dependently typed programming and proof
language, Beluga.

Meta-substitution for terms

[[C/X]]U (λx.M) = λx.M ′ where [[C/X]]U (M) = M ′

[[C/X]]U (u[σ]) = R′ where [[C/X]]U (σ) = σ′

and [[C/X]]U = [[Ψ̂.R/u]]P [Ψ]

and [σ′]Ψ(R) = R′

[[C/X]]U (u[σ]) = u[σ′] where [[C/X]]U (σ) = σ′

and [[C/X]]U 6= [[Ψ̂.R/u]]P [Ψ]

[[C/X]]U (x · S) = x · S′ where [[C/X]]U (S) = S′

[[C/X]]U (c · S) = c · S′ where [[C/X]]U (S) = S′

[[C/X]]U (p[σ] · S) = q[σ′′] · S′ where [[C/X]]U (σ) = σ′

and [[C/X]]U = [[Ψ̂.q[π]/p]]#A[Ψ]

and [σ′]Ψ(π) = σ′′

and [[C/X]]U (S) = S′

[[C/X]]U (p[σ] · S) = M ′ where [[C/X]]U (σ) = σ′

and [[C/X]]U = [[Ψ̂.x/p]]#A[Ψ]

and σ′Ψ(x) = M : α
and [[C/X]]U (S) = S′

and reduce(M : α, S′) = M ′

[[C/X]]U (p[σ] · S) = p[σ′] · S′ where [[C/X]]U (σ) = σ′ and
[[C/X]]U 6= [[Ψ̂.H/p]]#A[Ψ]

and [[C/X]]U (S) = S′

[[C/X]]U ((Φ̂. x)[s[σ]] · S) = M ′ where [[C/X]]U = [[Ψ̂.τ/s]]Φ[Ψ]

and [[C/X]]U (σ) = σ′

and [σ′]Ψ(τ) = τ ′

and τ ′Φ(x) = M : α
and [[C/X]]U (S) = S′

and reduce(M : α, S′) = M ′

[[C/X]]U ((Φ̂. x)[s[σ]] · S) = (Φ̂′. x)[s[σ′]] · S′ where [[C/X]]U 6= [[Ψ̂.τ/s]]Φ[Ψ]

and [[C/X]]U (Φ̂) = Φ̂′

and [[C/X]]U (σ) = σ′

and [[C/X]]U (S) = S′

Meta-substitution for substitutions

[[C/X]]U (·) = ·
[[C/X]]U (idψ) = σ where [[C/X]]U = [[Ψ/ψ]]G and

id(Ψ) = σ
[[C/X]]U (idψ) = idψ where [[C/X]]U 6= [[Ψ/ψ]]G
[[C/X]]U (σ,M) = σ′,M ′ where [[C/X]]U (σ) = σ′ and

[[C/X]]U (M) = M ′

[[C/X]]U (idψ[s[σ]]) = truncΦ(τ ′) where [[C/X]]U (σ) = σ′

and [[C/X]]U = [[Ψ̂.τ/s]]Φ[Ψ]

and [σ′]Ψ(τ) = τ ′

[[C/X]]U (idψ[s[σ]]) = sexpΨ(s[σ′]) where [[C/X]]U (σ) = σ′

and [[C/X]]U = [[Ψ/ψ]]G
[[C/X]]U (idψ[s[σ]]) = idψ[s[σ′]] where [[C/X]]U 6= [[Ψ/ψ]]G

and [[C/X]]U 6= [[Ψ̂.τ/s]]Φ[Ψ]

and [[C/X]]U (σ) = σ′

Meta-substitution for context

[[C/X]]U (·) = ·
[[C/X]]U (ψ) = Ψ where [[C/X]]U = [[Ψ/ψ]]G
[[C/X]]U (ψ) = ψ where [[C/X]]U 6= [[Ψ/ψ]]G
[[C/X]]U (Ψ, x:A) = Ψ′, x:A′ where [[C/X]]U (Ψ) = Ψ′ and

[[C/X]]U (A) = A′

Figure 5. Meta-substitution

In the future, we plan to extend our preliminary prototype to
fully support type reconstruction in the presence of substitution
variables and demonstrate their power in encoding strong normal-
ization proofs. We also plan to extend the coverage checker to guar-
antee that splitting over substitutions is defined on all cases. Finally,
we plan to address the issue of termination to guarantee that our
programs are total. This also requires us to check that a given re-
cursive data-type definition is strictly positive.

